Лабораторная работа №1

Математические модели динамики численности популяции одного вида

Мат. моделирование динамических процессов 1
БГУ, ММФ, 3 курс, 6 семестр
специальность Компьютерная математика и системный анализ
февраль 2025
доц. Лаврова О.А., доц. Щеглова Н.Л.

Задание 1. Модель Мальтуса

Модель Мальтуса, или модель экспоненциального роста является одной из простейших моделей численности популяции одного вида. Модель была предложена Мальтусом в 1798 году для описания роста численности населения Земли.

Содержательная постановка задачи

Рассмотрим популяцию больших размеров, численность которой измеряется миллионами и более. Полагаем, что не существует факторов, сдерживающих рост популяции, таких как болезни, хищники, конкурирующие виды, ограниченность питания и др. Пренебрегаем процессами иммиграции и эмиграции. Необходимо определить изменение численности популяции во времени.

Концептуальная постановка задачи

Численность популяции рассматривается как непрерывная функция времени $N = N(t)$. В силу того, что популяция достаточно велика, игнорируем случайные различия между отдельными особями. Полагаем, что каждая особь в популяции имеет равные шансы родить и умереть в течение определенного промежутка времени. Полагаем, что рождаемость и смертность непрерывны во времени и заданы коэффициентами рождаемости $\alpha(t) \geq 0$ и смертности $\beta(t) \geq 0$ на одну особь в единицу времени.

В основу модели Мальтуса положено утверждение, что скорость изменения численности популяции $N'(t)$ пропорциональна ее численности $N(t)$ в момент времени t, с коэффициентом пропорциональности, равным разности коэффициентов рождаемости и смертности

$$N'(t) = (\alpha(t) - \beta(t))\, N(t). \qquad (1)$$

Замечание

Предположение вида “скорость изменения величины пропорциональна значению самой величины (или некоторой функции от нее)” широко используется в различных областях знаний.

Математическая модель

Моделью Мальтуса называется задача Коши для уравнения (1) при заданном начальном условии в момент времени $t = t_0$

$$N'(t) = (\alpha(t) - \beta(t))\, N(t),$$
$$N(t_0) = N_0. \qquad (2)$$

Замечание

Модель Мальтуса является примером применения принципа аналогии при построении математических моделей объектов, для которых невозможно указать фундаментальные законы или вариационные принципы, которым они подчиняются.

Задание 1.1 (аналитическое решение)

Постройте аналитическое решение модели Мальтуса (2), например, методом разделения переменных для заданной численности населения N_0 в начальный момент времени t_0. **Сравните** построенное решение и решение, полученное с помощью функции **DSolve** (**DSolveValue**)

In[•]:=
```
ClearAll[t0, N0];
```

In[•]:=
```
sol = DSolveValue[{NN'[t] == (α[t] - β[t]) NN[t], NN[t0] == N0}, NN[t], t]
```

Out[•]=

$$e^{\int_1^t (\alpha[K[1]]-\beta[K[1]])\, dK[1] - \int_1^{t0} (\alpha[K[1]]-\beta[K[1]])\, dK[1]}\ N0$$

Напишите пользовательскую функцию для задания аналитического решения модели Мальтуса, где $\alpha(t)$, $\beta(t)$, $N_0 \geq 0$ и $t_0 \geq 0$ являются аргументами функции. Используйте решение, полученное Вами аналитически.

Задание 1.2 (график аналитического решения)

Изобразите в одной системе координат графики функции $N(t)$ для фиксированных значений t_0 и $N_0 > 0$ и различных значений $\alpha(t) = \alpha_0 = \text{const}$ и $\beta(t) = \beta_0 = \text{const}$. Рассмотрите случаи $\alpha_0 < \beta_0$, $\alpha_0 = \beta_0$, $\alpha_0 > \beta_0$. Используйте пользовательскую функцию из Задания 1.1 для задания функции $N(t)$.

Пример построения графиков решений

In[•]:=
```
t0 = 1; N0 = 10;
```

In[•]:=
```
funcs = (Activate@(sol /. #)) & /@
  {{α[_] → 0, β[_] → 2}, {α[_] → 2, β[_] → 2}, {α[_] → 2, β[_] → 0}}
```

Out[•]=

$\left\{10\, e^{-2\,(-1+t)},\ 10,\ 10\, e^{2\,(-1+t)}\right\}$

In[•]:=
```
Plot[funcs, {t, t0, 3}, PlotLegends → {"α0<β0", "α0=β0", "α0>β0"}]
```

Out[•]=



Задание 1.3 (качественный анализ модели по решению)

На основании построенных графиков в Задании 1.2 **сформулируйте** выводы о качественном поведении решения модели Мальтуса при $t \to \infty$ для случаев $\alpha_0 < \beta_0$, $\alpha_0 = \beta_0$, $\alpha_0 > \beta_0$: например, монотонное возрастание/убывание, экспоненциальное возрастание/убывание, колебания, экспоненциальный рост, гиперболический рост и т.д. На основании построенных графиков в Задании 1.2 **проанализируйте**, является ли положение равновесия $N(t) \equiv N_0 > 0$, соответствующее решению модели при совпадении коэффициентов рождаемости и смертности $\alpha_0 = \beta_0$, устойчивым относительно изменения значений параметров α_0, β_0? Результаты анализа запишите в текстовой ячейке.

Задание 1.4 (качественный анализ модели по фазовому портрету)

Фазовое пространство для рассматриваемой динамической системы первого порядка является одномерным и представляет собой прямую линию ON, которая ограничена значениями $N \geq 0$ в силу того, что $N(t)$ является численностью населения.

Изобразите фазовые траектории, например, с помощью функции **StreamPlot** решений динамической системы для случая постоянных коэффициентов $\alpha(t) = \alpha_0 = \text{const}$ и $\beta(t) = \beta_0 = \text{const}$ при $\alpha_0 < \beta_0$, $\alpha_0 = \beta_0$, $\alpha_0 > \beta_0$.
Пользуясь определением устойчивости в фазовом пространстве **проанализируйте**, являются ли положения равновесия динамической системы устойчивыми при условии, что $N(t) \geq 0$. Результаты анализа запишите в текстовой ячейке.

Например, для случая, когда рождаемость меньше смертности, т.е. $\alpha_0 < \beta_0$, имеем, что положение равновесия $N(t) \equiv 0$ является устойчивым.

```
In[•]:= α = 1; β = 2;

In[•]:= StreamPlot[{(α - β) x, 0}, {x, 0, 3}, {y, 0, .1}, AspectRatio → Automatic]

Out[•]=
```



Задание 2. Логистическая модель или модель Ферхюльста

Содержательная постановка задачи

Рассмотрим популяцию больших размеров, численность которой измеряется миллионами и более. Полагаем, что сдерживающим фактором роста популяции является ограниченность ресурсов. Пренебрегаем процессами иммиграции и эмиграции. Необходимо определить изменение численности популяции во времени.

Концептуальная постановка задачи и математическая модель

Логистическая модель или модель Ферхюльста (1838, 1845) принимает во внимание ограниченность доступных популяции ресурсов. В частности, считается, что существует равновесная численность популяции $N_p = \text{const} > 0$, которую может обеспечить окружающая среда, т.е. $N(t) \to N_p$ при $t \to \infty$. Это свойство называется эффектом насыщения численности населения.

При моделировании полагается также, что скорость изменения численности популяции пропорциональна самой численности с коэффициентом пропрциональности $k = \text{const} > 0$, дополнительно умноженной на величину отклонения численности от равновесного значения N_p

$$N'(t) = N(t)\, k \left(1 - \frac{N(t)}{N_p}\right), \qquad k = \text{const} > 0. \tag{3}$$

Логистической моделью называется задача Коши для уравнения (3) при заданном начальном условии в момент времени $t = t_0$

$$\begin{gathered} N'(t) = N(t)\, k \left(1 - \frac{N(t)}{N_p}\right), \\ N(t_0) = N_0. \end{gathered} \tag{4}$$

Замечания

Предположение о механизмах насыщения используются при построении многих моделей в различных областях знаний.

Существенным недостатком логистической модели является тот факт, что равновесная численность популяции N_p вводится в качестве известного параметра, в то время как нахождение этой величины нередко является основной задачей исследования.

Современные прогнозы показывают, что численность человечества в обозримом будущем стабилизируется на уровне $N_p = 12 * 10^9$ человек. Такой же прогноз дается ООН.

Задание 2.1 (аналитическое решение)

Постройте аналитическое решение логистической модели (4), например, методом разделения переменных для заданной численности населения N_0 в начальный момент времени t_0. **Сравните** построенное аналитическое решение и решение, полученное с помощью функции **DSolve (DSolveValue)**.
Напишите пользовательскую функцию для задания аналитического решения логистической модели, где $k > 0$, $N_p > 0$, $N_0 \geq 0$ и $t_0 \geq 0$ являются аргументами функции. Используйте решение, полученное Вами аналитически.

Задание 2.2 (график аналитического решения)

Изобразите в одной системе координат графики функции $N(t)$ для фиксированных значений k, N_p и t_0 и различных значений N_0. Рассмотрите случаи, когда $N_0 < N_p$, $N_0 = N_p$, $N_0 > N_p$. При выполнении задания используйте пользовательскую функцию из Задания 2.1 для описания аналитического решения логистической модели.

Задание 2.3 (качественный анализ модели по решению)

На основании построенных графиков в Задании 2.2 **сформулируйте** выводы о качественном поведении решения логистической модели при $t \to \infty$ для случаев $N_0 < N_p$, $N_0 = N_p$, $N_0 > N_p$ при $k > 0$.
На основании построенных графиков в Задании 2.2 **проанализируйте**, является ли положение равновесия $N(t) \equiv N_p$, соответствующее решению модели при $N_0 = N_p$, устойчивым по начальным данным? Результаты анализа запишите в текстовой ячейке.

Задание 2.4 (качественный анализ модели по фазовому портрету)

Фазовое пространство для рассматриваемой динамической системы первого порядка является одномерным и представляет собой прямую линию ON, которая ограничена значениями $N \geq 0$ в силу того, что $N(t)$ является численностью населения.

Изобразите фазовые траектории решений динамической системы, например, с помощью функции **StreamPlot**.
Пользуясь определением устойчивости в фазовом пространстве **проанализируйте**, являются ли положения равновесия динамической системы устойчивыми при условии, что $k > 0$, $N(t) \geq 0$. Результаты анализа запишите в текстовой ячейке.

Задание 2.5 (рост бактерий)

Предположим, что рост бактерий описывается логистической моделью. Известно, что в начальный момент времени $t = 0$ численность бактерий равна $N = N_p / 4$, в момент времени $t = 1$ численность $N = N_p / 2$. **Найдите**, в какой момент времени $N = 3\,N_p / 4$?

Задание 3. Нелинейный аналог модели Мальтуса

Содержательная постановка задачи

Рассмотрим популяцию больших размеров, численность которой измеряется миллионами и более. Полагаем, что не существует факторов, сдерживающих рост популяции, таких как болезни, хищники, конкурирующие виды, ограниченность питания и др. Полагаем также, что особи популяции заинтересованы в росте популяции. Пренебрегаем процессами иммиграции и эмиграции. Необходимо определить изменение численности популяции во времени.

Математическая модель

Полагаем, что коэффициент рождаемости пропорционален численности населения $\alpha(N) = \alpha_0\, N$, так как особи популяции заинтересованы в ее росте, где $\alpha_0 = \text{const} > 0$. Коэффициент смертности полагаем постоянным $\beta(t) = \beta_0 = \text{const} > 0$. Тогда уравнение Мальтуса (1) преобразуется к виду

$$N'(t) = N(t)\,(\alpha_0\, N(t) - \beta_0) \qquad (5)$$

с квадратичной нелинейностью в правой части.

Математической моделью называется задача Коши для уравнения (5) при заданном начальном условии в момент времени $t = t_0$

$$\begin{aligned} &N'(t) = N(t)\,(\alpha_0\, N(t) - \beta_0), \\ &N(t_0) = N_0. \end{aligned} \qquad (6)$$

Задание 3.1 (аналитическое решение)

Постройте аналитическое решение нелинейной модели (6) для заданной численности населения N_0 в начальный момент времени t_0. При построении аналитического решения необходимо отдельно рассматривать случаи, когда $N_0 < N_{кр}$, $N_0 = N_{кр}$, $N_0 > N_{кр}$, где $N_{кр} = \frac{\beta_0}{\alpha_0}$.
При $N_0 > N_{кр}$ решение строится только на отрезке $[t_0,\ t^*]$, где t^* необходимо найти из условия $\lim_{t \to t^*} N[t] = +\infty$.
Сформулируйте частный случай аналитического решения нелинейной модели (6) при условии, что $\beta_0 = 0$ ($N_{кр} = 0$).
Сравните построенное аналитическое решение нелинейной модели (6) и решение, полученное с помощью функции **DSolve (DSolveValue)**.

Напишите пользовательскую функцию для задания аналитического решения нелинейной модели Мальтуса, где $N_{кр} > 0$, $\beta_0 > 0$, $N_0 \geq 0$ и $t_0 \geq 0$ являются аргументами функции. Используйте решение, полученное Вами аналитически.

Задание 3.2 (график аналитического решения)

Изобразите в одной системе координат графики функции $N(t)$ для фиксированных значений α_0 и β_0 и различных значений N_0. Рассмотрите случаи $N_0 < N_{кр}$, $N_0 = N_{кр}$, $N_0 > N_{кр}$, где $N_{кр} = \frac{\beta_0}{\alpha_0}$. При выполнении задания используйте пользовательскую функцию из Задания 3.1 для описания аналитического решения нелинейной модели Мальтуса.

Задание 3.3 (качественный анализ модели по решению)

На основании построенных графиков в Задании 3.2 **сформулируйте** выводы о качественном поведении решения нелинейной модели при $t \to \infty$ для случаев $N_0 < N_{кр}$, $N_0 = N_{кр}$, $N_0 > N_{кр}$, $N_{кр} = \frac{\beta_0}{\alpha_0}$.
На основании построенных графиков в Задании 3.2 **проанализируйте**, является ли положение равновесия $N(t) \equiv N_{кр}$, соответствующее решению модели при $N_0 = N_{кр}$, устойчивым по начальным данным? Результаты анализа запишите в текстовой ячейке.

Задание 3.4 (качественный анализ модели по фазовому портрету)

Фазовое пространство для рассматриваемой динамической системы первого порядка является одномерным и представляет собой прямую линию ON, которая ограничена значениями $N \geq 0$ в силу того, что $N(t)$ является численностью населения.

Изобразите фазовые траектории решений динамической системы, например, с помощью функции **StreamPlot**.
Пользуясь определением устойчивости в фазовом пространстве **проанализируйте**, являются ли положения равновесия динамической системы устойчивыми при условии $N(t) \geq 0$, $\alpha_0 > 0$, $\beta_0 > 0$. Результаты анализа запишите в текстовой ячейке.

Задание 4. Предсказание численности населения Беларуси через 100 лет по заданным параметрам модели

На основе модели Мальтуса с постоянными коэффициентами из Задания 1 и на основе нелинейной модели из Задания 3 **сделайте предсказания** о численности населения Беларуси через 100 лет. Значения параметров модели α_0, β_0, N_0, t_0 определите, используя информацию из базы знаний Wolfram Alpha. Обратите внимание, что коэффициент рождаемости α_0, полученный из базы знаний Wolfram Alpha, должен быть изменен для использования в нелинейной модели Мальтуса: $\alpha_{0,\,\text{nonlinear}} = \alpha_0 / N_0$.

Сравните данные о численности населения Беларуси из базы знаний Wolfram Alpha с решением модели Мальтуса и с решением нелинейной модели Мальтуса.

Некоторые рекомендации по работе с базой знаний Wolfram Alpha

Создадим объект типа “Country”, ассоциированный с данными для имени “Belarus”

In[ ]:=
```
entityBelarus = Entity["Country", "Belarus"];
```

Получим необходимые свойства объекта entityBelarus и уберем размерности величин

In[ ]:=
```
{N0, α0, β0} = QuantityMagnitude@EntityValue[entityBelarus, #] & /@
   {"Population", "BirthRateFraction", "DeathRateFraction"}
```

Out[ ]=
```
{9 500 000, 0.0119, 0.0133}
```

Определим дату, которой соответствуют полученные данные

In[ ]:=
```
EntityValue[entityBelarus, #, "Date"] & /@
  {"Population", "BirthRateFraction", "DeathRateFraction"}
```

Out[ ]=
```
{Year: 2023, Year: 2017, Year: 2017}
```

In[ ]:=
```
t0 = 2017;
```

Скорректируем значение начальной популяции

In[ ]:=
```
N0 = QuantityMagnitude@Dated[entityBelarus, 2017]["Population"]
```

Out[ ]=
```
9 711 503
```

Построим график функции для всех значений свойства “Population”, которые хранятся в базе знаний

In[ ]:=
```
ts = Dated[entityBelarus, All]["Population"];
```

In[ ]:=
```
DateListPlot[ts, PlotRange → {{"1950", "2050"}, All}, Joined → False]
```

Out[ ]=



С помощью функций **TimeSeriesModelFit** и **TimeSeriesForecast** сделаем предсказание на следующие 100 лет по имеющимся данным из базы знаний WolframAlpha

In[∘]:=
```
tsModel = TimeSeriesModelFit[ts];
tsForecast = TimeSeriesForecast[tsModel, {100}];
```

··· TemporalData: The data is not uniformly spaced and will be automatically resampled to the resolution of the minimum time increment.

In[∘]:=
```
DateListPlot[{ts, tsForecast}, PlotRange → {{"1950", "2150"}, All}, Joined → False]
```

Out[∘]=



In[∘]:=
```
tsForecast["SliceData", "2125"]
```

Out[∘]=

{ 4.88654×10^6 people }

Преобразование данных к числовому формату для дальнейшего сравнения

In[∘]:=
```
tsData = Normal[ts];
tsData = MapAt[#["Year"] &, tsData, {All, 1}];
tsData = MapAt[Normal, tsData, {All, 2}];
```

In[∘]:=
```
ListPlot[tsData, PlotRange → {{1950, 2050}, All}]
```

Out[∘]=



Задание 5. Предсказание численности населения стран по историческим данным

По историческим данным о численности населения для заданной страны в период с 1950 года до 2010 года **сделайте предсказания** о значениях неизвестных параметров для

- модели Мальтуса с постоянными коэффициентами (см. Задание 1),
- логистической модели (см. Задание 2) и
- нелинейного аналога модели Мальтуса (см. Задание 3),

полагая, что $t_0 = 1950$ и N_0 являются заданными. Для этого необходимо аппроксимировать исторические данные функциями аналитических решений для указанных моделей с помощью функции **FindFit**.
Постройте графики относительной ошибки для найденных модельных решений в сравнении с реальными данными о численности населения.

Страны по вариантам:

1. Россия
2. Китай
3. Япония
4. Швеция
5. Франция
6. Греция
7. Марокко
8. Великобритания
9. Германия
10. Турция
11. Нидерланды
12. Бельгия
13. Катар
14. Египет

Рекомендации по выполнению Задания 5

Воспользуемся данными из базы знаний Wolfram Alpha для Беларуси

```
In[ ]:= data = {#, QuantityMagnitude@Dated[entityBelarus, #]["Population"]} & /@
    Range[1950, 2010];
```

In[•]:=
```
ListPlot[data]
```
Out[•]=



Параметры модели t_0 и N_0 полагаем заданными

In[•]:=
```
{t0, N0} = First@data
```
Out[•]=
```
{1950, 7722155}
```

Найдем неизвестные параметры для модели Мальтуса с использованием функции **FindFit**

In[•]:=
```
? FindFit
```
Out[•]=

Symbol

FindFit[*data*, *expr*, *pars*, *vars*] finds numerical values of the parameters *pars* that make *expr* give a best fit to *data* as a function of *vars*.

FindFit[*data*, {*expr*, *cons*}, *pars*, *vars*] finds a best fit subject to the parameter constraints *cons*.

In[•]:=
```
ClearAll[k, t]
```
In[•]:=
```
solutionMalthus = N0 Exp[k (t - t0)];
```
In[•]:=
```
FindFit[data, solutionMalthus, {k}, t]
```
Out[•]=
```
{k → 0.00593402}
```

Построим аналитическое решение для модели Мальтуса

In[•]:=
```
solutionMalthus = solutionMalthus /. FindFit[data, solutionMalthus, {k}, t]
```
Out[•]=

$7\,722\,155\, e^{0.00593402\,(-1950+t)}$

Построим график популяционной динамики на основании модели Мальтуса

In[∘]:=
```
Show[Plot[solutionMalthus, {t, 1950, 2010}], ListPlot[data]]
```

Out[∘]=



Построим график относительной ошибки для модели Мальтуса

In[∘]:=
```
errorMalthus = (Abs[Last@# - (solutionMalthus /. t → First@#) ] / Last@#) & /@ data;
```

In[∘]:=
```
{Norm[errorMalthus], Norm[errorMalthus, Infinity]}
```

Out[∘]=
```
{0.402473, 0.131303}
```

In[∘]:=
```
ListPlot[errorMalthus]
```

Out[∘]=



Задание 6. Предельный рост численности населения Земли

По данным о численности населения Земли, доступным из базы знаний Wolfram Alpha, **сделайте предсказание** о значении неизвестного параметра $\alpha_0 > 0$ для нелинейного аналога модели Мальтуса (см. Задание 3), полагая, что $\beta_0 = 0$, t_0 и N_0 являются заданными. Для нахождения неизвестного параметра модели α_0 необходимо аппроксимировать реальные данные функцией аналитического решения с помощью функции **FindFit**, см. Задание 5. При вызове функции **FindFit** укажите, что поиск значения α_0 осуществляется в окрестности 10^{-11}. Перед применение функции **FindFit** необходимо построить аналитическое решение для нелинейного аналога модели Мальтуса при $\beta_0 = 0$.

В рамках нелинейного аналога модели Мальтуса для найденного значения параметра α_0 **сделайте предсказание** о моменте времени t^*, при котором численность населения Земли примет бесконечное значение, т.е. $\lim_{t\to t^*} N[t] = +\infty$, см. формулу из Задания 3.1.

Изобразите данные о численности населения Земли из базы знаний Wolfram Alpha, а также модельное решение на временном отрезке $[t_0, t^*]$.

Постройте график относительной ошибки для найденного модельного решения в сравнении с реальными данными о численности населения Земли, см. Задание 5.

Рекомендации по выполнению Задания 6

Воспользуемся данными из базы знаний Wolfram Alpha о численности населения Земли

In[∘]:=
```
entityWorld = Entity["Country", "World"];
```

In[∘]:=
```
dataWorld = Dated[entityWorld, All]["Population"];
```

In[∘]:=
```
DateListPlot[dataWorld, Joined→False]
```

Out[∘]=



Преборазуем данные к числовому формату

In[∘]:=
```
dataWorld = Normal[dataWorld ];
dataWorld = MapAt[#["Year"] &, dataWorld , {All, 1}];
dataWorld = MapAt[Normal, dataWorld , {All, 2}];
```

In[•]:=
```
ListPlot[dataWorld, PlotRange → All, Frame → True, Axes → False]
```

Out[•]=



Параметры модели t_0 и N_0 полагаем заданными

In[•]:=
```
{t0, N0} = dataWorld⟦1⟧
```

Out[•]=

$\{-10\,000,\ 1. \times 10^6\}$

Задание 7* (необязательное). Математическая модель радиоактивного распада

Сформулируйте математическую модель радиоактивного распада вещества по аналогии с моделью Мальтуса из Задания 1. Для этого сделайте предположение, что скорость распада радиоактивного вещества пропорционален количеству этого вещества $x = x(t)$, где коэффициент пропорциональности является постоянным $k = \text{const} < 0$. Математическая модель формулируется в виде задачи Коши с начальным условием $x(t) = x_0$. Для более подробной информации смотрите [5, с. 18].

Используя вид аналитического решения математической модели, **найдите зависимость** между коэффициентом пропорциональности k и временем T полураспада радиоактивного вещества, когда $x(T) = x_0/2$?

Решите задачу 2.7 (определение времени почти полного распада вещества) и **задачу 2.8** (определение возраста горной породы) из [5, с.18].

Задание 8* (необязательное). Математическая модель распространения нового продукта на рынке

Данное задание является примером построения и анализа математической модели в экономике.

Сформулируйте математическую модель распространения нового продукта или услуги на рынке (модель Басса). В качестве неизвестной величины рассматривается количество пользователей $N(t)$ некоторого продукта или услуги на рынке. Используя принцип аналогии полагается, что за счет рекламы скорость изменения пользователей $N'(t)$ пропорциональна числу потенциальных клиентов $N_0 - N(t)$, где N_0 -- общее количество людей на рынке. Дополнительно в модели учитывается эффект косвенной рекламы за счет общения пользователей продукта с потенциальными клиентами. Для более подробной информации смотрите [1, с. 150].

Найдите момент времени, начиная с которого продолжать рекламу становится невыгодным. Для более подробной информации смотрите [1, с. 150].

Литература

[1] А. А. Самарский, А. П. Михайлов. Математическое моделирование: Идеи. Методы. Примеры. -- М.: Физматлит, 2001.

[2] А. С. Братусь, А. С. Новожилов, А. П. Платонов. Динамические системы и модели биологии. -- 2011.

[3] Д. Мюррей. Математическая биология. Том I. Введение. -- М.-Ижевск, 2009.

[4] B. Barnes and G. R. Fulford. Mathematical Modelling with Case Studies: A differential equation approach using Maple and MATLAB. Second Edition. -- CRC Press, 2008.

[5] Р. А. Прохорова. Обыкновенные дифференциальные уравнения : учебное пособие. -- Мн. : БГУ, 2017. http://elib.bsu.by/handle/123456789/205697

[6] В. В. Амелькин. Дифференциальные уравнения : учебное пособие. -- Мн. : БГУ, 2012. http://elib.bsu.by/handle/123456789/43871